
* Juicio político, cultura legal y escudo popular
* Anibal Perez-Linan
* Revista SAAP
*  Vol. 10, Nº 1, junio 2016, 11-33

use "SAAP-ECU.dta", clear

mvdecode _all, mv(88 99)

/*    VARIABLES    */

/*** Escudo popular ***/
/* frente al Congreso */
tab P52
recode P52  (1 2 = 1) (3 4 = 0) (missing = 0) , gen(Escudo) 
label var Escudo "Escudo al presidente"
tab Escudo


/*** Proximidad politica ***/
/* Confianza en el presidente - Confianza en Asamblea */
gen propres = P16G-P16Q
recode propres   (missing = 0)
label var propres "Confianza: Asamblea (-10) - Presidente (+10)"
/* Apoyo al presidente frente a la Asamblea */
recode P54 (1 2 = 1) (3 4 = 0) (missing = 0) , gen(P54rec)  
label var P54rec "Malo que Asamblea se oponga al Presidente"
/* Resultados */
recode P1 (3 = 1) (1 2 = 0) (missing = 0) , gen(P1rec) 
label var P1rec "Situación del país es mejor"


/*** Cultura legal  ***/
*tab   P53 /* Apoyo al presidente frente al Poder Judicial */
recode P53  (1 = 1) (2 = 0) (missing = 0) , gen(P53rec)
label var P53rec "Presidente puede ignorar juez"
label define  P53recL 0 "Presidente: No puede ignorar a un juez" 1 "Puede ignorar", add
label values P53rec P53recL

*tab   P29 /* Pueblo debe obedecer leyes, puede cambiar, o puede desobedecer */
recode P29 (2 = 1) (1 3 = 0) (missing = 0) , gen(P29rec) 
label var P29rec "Pueblo puede cambiar las leyes"
label define  P29recL 0 "Pueblo puede cambiar leyes: No" 1 "Si", add
label values P29rec P29recL

*tab   P46C /* Hay veces que es correcto desobedecer al gobierno */
recode P46C (3 4 = 1) (1 2 =0) (missing = 0) , gen(P46Crec) 
label var P46Crec "Es correcto desobedecer al gobierno"
label define  P46CrecL 0 "Desobedecer: No es correcto" 1 "Correcto", add
label values P46Crec P46CrecL

*tab   P9 /* Tradicional, modo de vida  */
recode P9 (3 4 = 1) (1 2 = 0) (missing = 0) , gen(P9rec)
label var P9rec "Cultura legal tradicional"
label define  P9recL 0 "Cultura legal: No-tradicional" 1 "Tradicional", add
label values P9rec P9recL

*tab   P47 /* De acuerdo con pena de muerte */
recode P47 (3 4 = 1) (1 2 =0) (missing = 0) , gen(P47rec) 
label var P47rec "De acuerdo con pena de muerte"
label define  P47recL 0 "Pena de muerte: En desacuerdo" 1 "De acuerdo", add
label values P47rec P47recL

/* Disciplina civica */
tab    P7D   /* Cuando usted piensa que tiene la razón, ¿está dispuesto a ir en contra de la ley?  (No) */ 
recode P7D  (1 = 1) (2 3 = 0)   (missing = 0) , gen(disc1)
tab    P13 /* ¿Si una decisión fue tomada por la mayoría de las personas, usted..? (obedece aunque no le guste) */
recode P13 (1 = 1) (2 = 0)   (missing = 0) , gen(disc2)
tab    P22A  /* Hay momentos que hay que desobedecer a la ley  ( desacuerdo/ muy en desacuerdo)*/ 
recode P22A (1 2 = 1) (3 4 = 0)   (missing = 0) , gen(disc3)
tab    P29   /*El pueblo debe obedecer siempre las leyes (siempre) */
recode P29 (1 = 1) (2 3 = 0)   (missing = 0) , gen(disc4)
tab P46B  /* La gente debería obedecer las leyes hechas por la Asamblea, aún si están en contra de su manera de pensar (de acuerdo / muy ) */
recode P46B (3 4 = 1) (1 2 = 0)   (missing = 0) , gen(disc5)
gen disciv = disc1+disc2+disc3+disc4+disc5
recode disciv (0 1 2 = 0) (3 4 5 = 1) (missing = 0), gen(disciv2)

/* Controles */
recode SEXO (1 = 0) (2 = 1) (missing = 0), gen(mujer)
label var mujer "Sexo (mujer)"
label var P72B "Años de educación"
recode P75A (1 = 1) (2 = 0) (missing = 0), gen(P75Arec) /* piso tierra */
label var P75Arec "Piso de tierra"
recode P75J (1 = 1) (2 = 0) (missing = 0), gen(P75Jrec) /* internet */
label var P75Jrec "Internet en casa"
recode P78 (missing = 7)
label var P78 "Etnicidad"


/*  TABLA 1  */	

logit        Escudo   i.P53rec                           /// cultura legal 
                      propres P54rec P1rec               /// proximidad politica
					  P29rec P46Crec P9rec P47rec        /// cultura 
					  i.CIU i.mujer P72B i.P75Arec  i.P75Jrec i.P78 


/* FIGURA 1 */	
					  
margins i.P53rec, at (propres=(-10(1)10))  
marginsplot, recast(line) recastci(rarea)  

					  
/* PSM & TABLA 2  */

probit  P53rec        c.propres P54rec P1rec             /// proximidad politica
					  P46Crec P47rec  					  

teffects psmatch (Escudo) (P53rec                        /// 
                      c.propres P54rec P1rec             /// proximidad politica
					  P46Crec P47rec                     /// cultura 
					  , probit),  generate(nneigh)			  			  
					  
tebalance summarize propres P54rec P1rec                 /// proximidad politica
					        P46Crec P47rec					  
					  
